# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=hls-test-utils
pkgname=haskell-hls-test-utils
pkgver=1.5.0.0
pkgrel=99
pkgdesc="Utilities used in the tests of Haskell Language Server"
url="https://github.com/haskell/haskell-language-server#readme"
license=("Apache")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-aeson' 'haskell-async' 'haskell-blaze-markup' 'haskell-data-default'
         'haskell-extra' 'haskell-ghcide' 'haskell-hls-graph' 'haskell-hls-plugin-api'
         'haskell-lens' 'haskell-lsp' 'haskell-lsp-test' 'haskell-lsp-types' 'haskell-tasty'
         'haskell-tasty-expected-failure' 'haskell-tasty-golden' 'haskell-tasty-hunit'
         'haskell-tasty-rerun' 'haskell-temporary' 'haskell-unordered-containers')
makedepends=('ghc' 'uusi')
source=("https://hackage.haskell.org/packages/archive/$_hkgname/$pkgver/$_hkgname-$pkgver.tar.gz")
sha512sums=('e58d811452ae48a86e5a8a1621aedd2a1fec39f33aec30a1cf62c0a6782d4de72c8db2a206ef55b67bc00694eb12a0d83b0a5424f4852712700364f9eb1cec7b')

prepare() {
  cd $_hkgname-$pkgver
  gen-setup
}

build() {
  cd $_hkgname-$pkgver

  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
    --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
    --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
    --ghc-option=-optl-Wl\,-z\,relro\,-z\,now \
    --ghc-option='-pie'

  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $_hkgname-$pkgver
  runhaskell Setup test --show-details=direct
}

package() {
  cd $_hkgname-$pkgver

  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
  install -D -m644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname/
  rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
